Institut für Informatik WS 2005/2006 der Universität München Prof. M. Hofmann A. Abel 1. Dezember 2005

Übungen zur Vorlesung Rechnergestütztes Beweisen

Blatt 7

Aufgabe 17: [Papier, 5 Punkte] Betrachten Sie die Theorie der Paare und Projektionen. In der Sprache L gebes einen Grundtyp β. Für je zwei Typen τ_1 und τ_2 gebe es den Produkttyp τ_1 ∗ τ_2 und 3 Funktionssymbole: Paarbildung ⟨ , ⟩ : [τ_1, τ_2 → τ_1∗τ_2] (Mixfix-Schreibweise), erste Projektion .1 : [τ_1×τ_2 → τ_1] und zweite Projektion .2 : [τ_1 ×τ_2 → τ_2] (Postfix-Schreibweise). Die Gleichungstheorie wird erzeugt von den Axiomen

⟨x, y⟩.1 = x ⟨x, y⟩.2 = y ⟨x.1, x.2⟩ = x .

Im Folgenden bezeichne T_τ die Terme vom Typ τ und N_τ die neutralen Terme vom Typ τ, das sind Folgen von Projektionen angewandt auf eine Variable. Die Theorie hat das folgende Modell: [[β]] = N_β, [[τ_1 ∗ τ_2]] = [[τ_1]]×[[τ_2]] (kartesisches Produkt), [[⟨ , ⟩]](u, v) = (u, v), [[.1]](u, v) = u, [[.2]](u, v) = v. Nun definieren wir:

q_τ : [[τ]] → T_τ q_β(n) = n q_{τ1∗τ2} (u, v) = ⟨q_{τ1} (u), q_{τ2} (v)⟩

u_τ : N_τ → [[τ]] u_β(n) = n u_{τ1∗τ2} (n) = (u_{τ1} (n.1), u_{τ2} (n.2)) Die Identitätsbelegung ρ_0 sei definiert durch ρ_0(x) = u_τ(x) für jede Variable x : τ. Weisen Sie nach, dass

σ_τ : T_τ → T_τ σ_τ(t) = q_τ([[t]]_{ρ0} )

ein Kanonisierer ist und berechnen Sie σ_{τ1∗τ2} (⟨x, y⟩.1).

Aufgabe 18: (Merge-Sort) [PVS, 15 Punkte] Zeigen Sie die Korrektheit von Merge-Sort. Von den drei beteiligten Funktionen split, merge und sort ist nur die erste primitiv rekursiv, d.h. der rekursive Aufruf erfolgt auf der unmittelbaren Vorgänger-Liste cdr(l). Deswegen sind nur die Beweise über split durch gewöhnliche Listen-Induktion bequem machbar. Für merge und
sort definiert man sich zuerst passende Induktionsschemata merge induct

und sort induct, die den Rekursionsschemata der jeweiligen Funktionen entsprechen. Dabei ist das erste Schema durch eine doppelte Induktion, und das zweite Schema durch Maß-Induktion (measure-induct+ ...) zu beweisen. Eine Aussage über merge beweist man dann bequem mit merge induct , und analog für sort .

Sortierung spezifiziert man primitiv rekursiv mit Hilfe der Hilfsrelation lb. Dabei gelte lb(x,l), falls das Element x kleiner gleich jedem Element der Liste

l ist. Für sortiertes l ist dieser Test entsprechend einfach zu implementieren.

Führen Sie die Beweise untenstehender Aussagen in PVS und denken Sie auch an die TCCs.

mergesort : THEORY
  BEGIN

    t: TYPE
    x,y,z : VAR t
    l,l1,l2,xs,ys,zs : VAR list[t]

    % cases on lists
    list_case : LEMMA
        FORALL (p:[list[t] -> boolean]): FORALL (l):
          (l=null IMPLIES p(null)) AND
          (FORALL(x,xs):l=cons(x,xs) IMPLIES p(cons(x,xs))) IMPLIES
                p(l)

    % less than

    lt : [t,t->bool]
    lt_ax : AXIOM
      FORALL (x,y:t): NOT (lt(x,y) AND lt(y,x))

    % splitting

    split(l:list[t],flag:boolean) : RECURSIVE [list[t],list[t]] =
       IF null?(l)
         THEN (null,null)
         ELSE LET (u,v) = split(cdr(l),NOT flag) IN
           IF flag THEN (cons(car(l),u),v)
             ELSE (u,cons(car(l),v))
           ENDIF
      ENDIF

MEASURE length(l)

split1 : LEMMA
  FORALL(l:list[t], flag:boolean):
    LET (u,v) = split(l,flag) IN
      length(u) <= length(l) AND
      length(v) <= length(l) AND
      (NOT null?(l) AND NOT null?(cdr(l)) IMPLIES
        length(u) < length(l) AND
        length(v) < length(l))

% merging

merge_induct : LEMMA
  FORALL (p : [list[t],list[t] -> boolean]):
    (FORALL (l2): p(null,l2)) AND
    (FORALL (l1): p(l1,null)) AND
    (FORALL (x,y,xs,ys): p(xs,cons(y,ys)) AND p(cons(x,xs),ys)
      IMPLIES p(cons(x,xs),cons(y,ys)))
    IMPLIES (FORALL (l1,l2): p(l1,l2))
% induction on l1 and l2

merge(u:list[t], v:list[t]) : RECURSIVE list[t] =
  IF null?(u) THEN v
  ELSIF null?(v) THEN u
  ELSIF lt(car(u),car(v))
    THEN cons(car(u),merge(cdr(u),v))
    ELSE cons(car(v),merge(u,cdr(v)))
  ENDIF
MEASURE length(u)+length(v)

% merge sort

sort_induct : LEMMA
  FORALL (p : [list[t] -> boolean]):
    p(null) AND
    (FORALL (x): p (cons(x,null))) AND
    (FORALL (x,y,l): p(split(cons(x,cons(y,l)),TRUE)‘1) AND
                     p(split(cons(x,cons(y,l)),TRUE)‘2) IMPLIES
                     p(cons(x,cons(y,l))))
    IMPLIES FORALL (l): p(l)
% measure-induct

sort(l:list[t]) : RECURSIVE list[t] =

  IF null?(l)
    THEN null
  ELSIF null?(cdr(l))
    THEN l
    ELSE LET (u,v) = split(l,TRUE) IN
      merge(sort(u),sort(v))
  ENDIF
MEASURE length(l)

% occurrences and permutation

occ(x:t,l:list[t]) : RECURSIVE nat =
  CASES l
    OF null : 0 ,
    cons(y,l) : IF x=y THEN occ(x,l)+1 ELSE occ(x,l) ENDIF
  ENDCASES
MEASURE length(l)

perm(l1,l2 : list[t]) : bool = FORALL(x:t):occ(x,l1) = occ(x,l2)

split2 : LEMMA
   FORALL(l:list[t],flag:boolean):
     LET (u,v) = split(l,flag) IN
       FORALL(x:t):occ(x,l) = occ(x,u) + occ(x,v)

merge1 : LEMMA
   FORALL(u,v:list[t]):
       FORALL(x:t):occ(x,merge(u,v)) = occ(x,u)+occ(x,v)
% use merge_induct

% lower bound of a sorted list

lb (x:t, l:list[t]) : boolean =
  null?(l) OR NOT lt(car(l),x)

% "sorted" in terms of "lower bound"

sorted(l:list[t]) : RECURSIVE boolean =
  IF null?(l)
     THEN TRUE
     ELSE sorted(cdr(l)) AND lb(car(l),cdr(l))
  ENDIF
MEASURE length(l)

  % merge preserves lower bounds

  lb_merge : LEMMA
      FORALL (x): FORALL (l1,l2):
      lb(x,l1) AND lb(x,l2) IMPLIES lb(x,merge(l1,l2))
  % use merge_induct

  % merge preserves sortation

  merge2 : LEMMA
     FORALL(u,v:list[t]):
         sorted(u) AND sorted(v) IMPLIES sorted(merge(u,v))
  % use merge_induct and lb_merge

  sort_correct : THEOREM
       FORALL(l:list[t]):
          sorted(sort(l)) AND perm(l,sort(l))
  % use sort_induct, merge1 and merge2

END mergesort

Abgabe: bis Dienstag, 06.12., 18.00 Uhr im Sekretariat (Z1.05) oder bei Andreas Abel im Büro (D1.09). Elektronische Lösungsteile per Mail an

abel@informatik.uni-muenchen.de.